



Implementation Guide for Administrators

Creating a Balance Sheet and Cash Flow Statement

Copyright 2011 Adaptive Planning, Inc. All Rights Reserved. This work contains trade secrets and confidential material of Adaptive Planning, Inc., and its use or disclosure in whole or in part without the express written permission of Adaptive Planning, Inc., is prohibited.

Last Updated October, 2011

**Adaptive Planning, Inc.
Balance Sheet/Cash Flow Guide**

Table of Contents

1. INTRODUCTION	1
HOW MANY ENTITIES?.....	1
2. CREATE THE ACCOUNTS	2
BALANCE SHEET GL ACCOUNTS	2
<i>How Many Accounts Needed?.....</i>	<i>2</i>
<i>Planned by Monthly Delta or Balance</i>	<i>3</i>
BS AUDIT METRIC ACCOUNT	4
CASH FLOW CUSTOM ACCOUNTS.....	5
<i>Weight the Beginning Cash Account.....</i>	<i>6</i>
3. CREATE THE SHEETS	8
BALANCE SHEET	8
CASH FLOW STATEMENT SHEET	9
4. IMPORT ACTUAL BALANCE SHEET DATA.....	10
DOES ACTUAL DATA BALANCE?	10
5. WRITE FORMULAS.....	11
STEPS TO FOLLOW.....	11
HINTS	11
<i>Rollup = Yes.....</i>	<i>11</i>
<i>Multiple Entities</i>	<i>12</i>
<i>BS Accounts with No Formulas.....</i>	<i>12</i>
<i>BS Drivers.....</i>	<i>13</i>
Custom Account or Global Assumption?.....	13
YTD EARNINGS AND RETAINED EARNINGS	14
BALANCE THE BALANCE SHEET	15
<i>One Account.....</i>	<i>15</i>
<i>Two Accounts</i>	<i>17</i>
<i>Three Accounts</i>	<i>18</i>
<i>A Different Approach</i>	<i>19</i>
ACCOUNTS RECEIVABLE	20
<i>Allowance for Doubtful AR</i>	<i>21</i>
PREPAID EXPENSES	22
<i>Use a Modeled Sheet</i>	<i>22</i>
<i>Plan the Prepaid Balance Sheet Accounts</i>	<i>23</i>
<i>Create a Prepaid Audit Sheet.....</i>	<i>23</i>
FIXED ASSETS	24
<i>Accumulated Depreciation</i>	<i>24</i>
ACCOUNTS PAYABLE.....	25
OTHER PAYABLES.....	26
DEFERRED REVENUE	27
DEFERRED INCOME TAX	28
OTHER DEFERRALS	29
<i>How to Accumulate a Revenue or Expense Account</i>	<i>29</i>

**Adaptive Planning, Inc.
Balance Sheet/Cash Flow Guide**

Table of Contents

6.	BALANCE SHEET ISSUES WITH MULTIPLE CURRENCY	31
7.	CASH FLOW STATEMENT	35
	Net Income (Loss).....	35
	Net Income Adjustments.....	35
	Change in Operating Assets & Liabilities	35
	Cash Flow from Investing	36
	Cash Flow from Financing	36
	TROUBLESHOOTING THE CASH FLOW STATEMENT	37
8.	APPENDICES	38
	APPENDIX A: CASH FLOW ACCOUNT IMPORT.....	38
	APPENDIX B: CTA CALCULATION EXAMPLE.....	39

Adaptive Planning, Inc.

Balance Sheet/Cash Flow Guide

1. Introduction

The process of developing a Balance Sheet and Cash Flow Statement involves creating GL balance sheet accounts and custom cash flow accounts, putting them on sheets, and writing formulas in them.

Balance sheets are typically formulated to reference P&L accounts, plus sometimes drivers such as Days Sales Outstanding (in the case of Accounts Receivable.) The cash flow statement then summarizes the changes in balance sheet accounts. The P&L, Balance Sheet, and Cash Flow Statement comprise a set of fully integrated financial statements. Audit mechanisms can be put in place to ensure that the three statements are properly integrated.

The cash flow statement is usually designed to be entirely formulated, with no data entry. While it is possible to have the cash flow statement as the data entry point, driving the change in balance sheet accounts accordingly, this guide presumes that the balance sheet drives the cash flow statement.

This guide describes a suggested path to follow in creating a Balance Sheet and Cash Flow Statement, with step-by-step instructions and specific examples.

How Many Entities?

Before getting started, consider how many plans require balance sheets and cash flow statements. If only one BS/CF is needed, this is typically and most efficiently created at the top of the plan tree, at the Corporate Plan level. But if the plan tree includes multiple key entities or business units, a BS/CF may be required at each such subplan level. These subplan BS/CFs will roll up to the total company BS/CF.

If multiple BS/CFs are required, the formulas will need to be written in multiple plans. Strategies for implementing this are addressed in the guide below.


Adaptive Planning, Inc. Balance Sheet/Cash Flow Guide

2. Create the Accounts

Ideally, the GL balance sheet accounts (Asset, Liabilities, and Equity) were imported during the initial structure setup. If they were not, they should be added to the GL account tree now.

Balance Sheet GL Accounts

The GL balance sheet “root” account categories are standard in the Adaptive Planning application. (In other words, they are created automatically when a company model is created.) These categories are Assets (Current Assets, Fixed Assets, Long Term Assets, Other Assets); Liabilities and Equities (Liabilities, Current Liabilities, Long Term Liabilities, Equity); and Equity Net Income.




How Many Accounts Needed?

There are often many more balance sheet accounts used in the General Ledger than are necessary for planning. If this is the case, a shorter subset can be created in Adaptive Planning. Upon initial import of actual balance sheet data, the many GL accounts can be mapped to the fewer Adaptive Planning accounts. Or, all the accounts can be set up in Adaptive Planning. In this case, the key accounts (e.g. cash, accounts receivable, accounts payable) can be planned with formulas, and the rest can simply carry balances forward from the last month of actual data. Either method is acceptable.

Adaptive Planning, Inc. Balance Sheet/Cash Flow Guide

Planned by Monthly Delta or Balance

Balance sheet accounts are cumulative accounts, and can be planned by monthly delta, or balance. At the beginning, they should all be set to the default, planned by monthly delta.



The screenshot shows a form titled "Account Details" with the following fields:

- * Code: 1300_01
- * Name: 1300_01 Hardware
- Type: Fixed Asset (cumulative)
- Rolls up to: 1300_00 Property & Equipment
- Planned by: Monthly Delta Balance
- Display as: Currency

The "Planned by" section is circled in red, highlighting the radio buttons for "Monthly Delta" and "Balance".

As balance sheet modeling develops, some accounts may need to be changed to be “planned by balance.” Following are examples of the two different choices:

Planned by monthly delta: The account’s ending balance is calculated as beginning balance + inputs – outputs = ending balance. E.g. Accounts Receivable is calculated as beginning balance + new billings – payments from customers = ending balance.

Planned by balance: The account’s ending balance is entered or calculated with a formula, e.g. Accounts Receivable balance is calculated as an assumed DSO (e.g. 30 days) x average daily sales.

Accounts which are planned by balance can contain splits. Accounts which are planned by Monthly Delta cannot.

→ Again, at this point, **all balance sheet accounts should be created with the default choice, Plan by Monthly Delta.** This can later be changed for specific accounts, as needed.

Adaptive Planning, Inc. Balance Sheet/Cash Flow Guide

BS Audit Metric Account

Create a **metric account** to audit the balance sheet. Its formula should equal Assets minus Liabilities and Equity. This account will be placed at the bottom of the Balance Sheet. Its balance should always be zero. (More on this, below.)

Account Details

* Code:

* Name:

Rolls up to:

Display as:


Decimal places:

* Formula:

Adaptive Planning, Inc. Balance Sheet/Cash Flow Guide

Cash Flow Custom Accounts

Create a custom account for each row on the Cash Flow Statement. The format is fairly standard from company to company. It begins with cash flow from operations, followed by cash flow from investments and financing activities. At the end is a summary of Beginning Cash +/- Change in Cash = Ending Cash. This ending cash calculated here is compared to ending cash on the balance sheet.



Adaptive Planning, Inc. Balance Sheet/Cash Flow Guide

Go to Administration, "Manage your Custom Accounts," and set up the necessary accounts.

→ **Hint:** Begin by creating a group called Cash Flow Custom Accounts. Add parent and sub-accounts in the same rollup hierarchy as on the cash flow statement. Use a naming convention, such as beginning all codes with "CF," for example, to facilitate identifying these on sheets and in formulas.

Please see **Appendix A** for an example of an account import template for Cash Flow accounts.

	A	B	C	D
1				
2				
3	Required			
4	Name	Rolls Up To	Type	Code
5	Cash Flow	Custom		
6	CF from All Sources	Cash Flow	Custom (periodic)	CF_from_All_Sources
7	Cash Flow from Operating Activities	CF from All Sources	Custom (periodic)	CF_from_Ops
8	Net Income (Loss)	Cash Flow from Operating Activities	Custom (periodic)	CF_NI
9	Net Income Adjustments	Cash Flow from Operating Activities	Custom (periodic)	CF_NI_Adj
10	Depreciation & Amortization	Net Income Adjustments	Custom (periodic)	CF_Deprec_Amort
11	Deferred Taxes	Net Income Adjustments	Custom (periodic)	CF_Def_Tx
12	Change in Operating Assets & Liabilities	Cash Flow from Operating Activities	Custom (periodic)	CF_Chg_Oper_A_L

Weight the Beginning Cash Account

On a Cash Flow Statement viewed at a time period rollup, such as Q1, Beginning Cash for the time period rollup should be the beginning cash for the first month of the quarter or year. Change in Cash should be the sum of the months in the quarter or year. Ending Cash should be the ending balance in the last month of the quarter or year.

Change in Cash will behave properly if it is set up as a periodic account (as indicated on the attached import file.) **Ending Cash** will also behave properly if it is set up as a cumulative account planned by balance (as indicated on the attached import file.)

However, **Beginning Cash** needs some attention before it will behave as desired. The following steps will result in Beginning Cash displaying beginning cash from the first month of the quarter or year:

1. Create a new Assumption called Beginning Cash Weight Acct and add this to the Assumptions sheet or another plan independent sheet.

Account Details

Assumption

* Code:

* Name:

Type: Periodic Link

Time rollup:

Rolls up to:

Display as:

Decimal places:

Actuals: Enable actuals
 No actuals for account (show plan data)

Description:

Adaptive Planning, Inc. Balance Sheet/Cash Flow Guide

- On the Assumptions sheet, create a formula that populates 1,000,000,000,000 for January and 1 for April, July, and October, using the following formula:

=if(month(this)=1, 1000000000000, if(month(this)=4,1, if(month(this)=7,1, if(month(this)=10,1,0))))

→ **Note:** Be sure to put the formula in both Actual and Plan versions.

- In the Beginning Cash custom account, set the Time rollup to Weighted average of rolled-up values, and choose as the weight the Assumption created in the previous step.

Account Details

* Code:

* Name:

Type: Periodic Cumulative Link

Planned by: Monthly delta Balance

Time rollup:

Weighted by: [Edit](#)

Plan/Dimension rollup:

Rolls up to:

Display as:

Now, when the Cash Flow Statement displays quarterly or annual time period rollups, the Beginning Cash row will be accurate:

	Jan	Feb	Mar	Q1
Net Increase (Decrease) in Cash	-27,088,546	-4,105,070	-23,520,629	-54,714,245
Cash at Beginning of Period	215,275,803	188,187,258	184,082,188	215,275,803
Cash at End of Period	188,187,257	184,082,188	160,561,559	160,561,559

Adaptive Planning, Inc. Balance Sheet/Cash Flow Guide


3. Create the Sheets

Balance Sheet

Once the balance sheet accounts have been established, create a standard sheet to display them. Formulas will be built using this sheet.


Go to Admin, Manage Sheets on Plans, and create a new standard sheet.

Name the sheet "Balance Sheet." Do not check the "Show Initial Balance" box. (Imported actual data will be used to create initial balances for planning purposes.)



Select sheet type: Standard
Sheet name: Balance Sheet
Next > Cancel

Create three account groups on the sheet. Add the Assets account rollup to the first group; the Liabilities and Equities account rollup to the second group; and the metric account called BS Check to the third group.



Edit Sheet Account Groups

- New Group
 - Assets (1)
 - Current Assets
 - Fixed Assets
 - Other Assets
- New Group
 - Liabilities and Equities (2)
 - Liabilities
 - Equity
- New Group
 - BS Check (3)

A fourth group may be added later, if custom accounts are used to hold balance sheet drivers, such as DSO.

Adaptive Planning, Inc. Balance Sheet/Cash Flow Guide

Make the sheet available to the plans that are going to have BS/CF, most commonly only the Corporate Plan. (Or, if more than one balance sheet is to be planned, e.g. one for US and one for Europe, choose the appropriate rollup departments.)


Using the top of the plan tree (Corporate Plan) is more efficient than using a special balance sheet planning department, e.g. 90-000 Balance Sheet. This is because the Corporate Plan department has access to all of the data in the plan, and this is necessary to drive the balance sheet. A separate, standalone department has access to no data except the data in that department. Accounts needed to drive the balance sheet would have to be made public to capture the data necessary to drive the balance sheet (e.g. total company revenue drives AR; total company expenses drive AP.) This is possible, but it is more efficient to plan the balance sheet at the top of the plan tree.

Cash Flow Statement Sheet


Once the cash flow accounts have been established, create a standard sheet to display them. Formulas will be built using this sheet.

Go to Admin, Manage Sheets on Plans, and create a new standard sheet.

Name the sheet "Cash Flow Statement."



Add the custom account group called "Cash Flow Custom Accts." Do not check the "Show Initial Balance" box.



Make the sheet available to the same plan(s) which have access to the Balance Sheet.

Adaptive Planning, Inc.

Balance Sheet/Cash Flow Guide

4. Import Actual Balance Sheet Data

It can be helpful to import at least one month of actual balance sheet data at this point. Seeing the actual balance sheet data can offer clues regarding both the complexity of the balance sheet, as well as which accounts need to be planned.

This guide assumes that both Assets and Liabilities/Equity are imported and planned as positive numbers, in the same way that both Income and Expenses are imported and planned as positive numbers. (This is true for the most part. There are obvious exceptions, such as Sales Returns and Accumulated Depreciation, which have normal negative balances.)

The calculations described below are based on this assumption. Liabilities/equity can be displayed as negative numbers, but this will affect some of the calculations below, as noted. If liability and equity accounts are imported negative numbers, some of the formulas below must be reversed.

If possible, import actual balance sheet data only at the same plan levels that are to be used in planning balance sheet/cash flow. For example, if the company is to have only one BS/CF, at the Corporate Plan level, try to import the actual BS data into the Corporate Plan (only), rather than into individual subplans, which roll up to the Corporate Plan. Having all of the actual data at the Corporate Plan (Only) level will mean that this data is all at the same level as the BS/CF formulas. The total BS, actual data and planned formulas, can be viewed at Corporate Plan (Only), without having to switch to Corporate Plan (Rollup).

If the actual data needs to be imported at multiple subplan levels, even though formulas are written only at the Corporate Plan level, the sheet will need to be viewed at Rollups level in order to see all data and resulting calculations.

Does Actual Data Balance?

Import at least the last month of actual data. Once it has been imported, open the Balance Sheet and see if the actual data balances.

If it does not balance, this may be due to **YTD Earnings**. Check the Equity section and see if any YTD Earnings data has been imported. If it has not, a formula should be written in the Actuals version, in the YTD Earnings account, to pull net income from the P&L. (For more information on this topic, Please see the section below called YTD Earnings and Retained Earnings.)

If the balance sheet still does not balance after the YTD Earnings formula has been written, the source data should be investigated. A balance sheet that balances is an important starting point.

Adaptive Planning, Inc. Balance Sheet/Cash Flow Guide

5. Write Formulas

Steps to Follow

This guide recommends the following steps for creating balance sheet and cash flow formulas:

- Write the formulas using the Balance Sheet and Cash Flow sheets, not the Formulas tab. This allows for most efficient formula trouble-shooting. View the sheets at the Corporate Plan (or other parent plan) level, in Edit Plan mode.



- Write each balance sheet formula first (e.g. in Accounts Receivable), and then write the corresponding formula on the Cash Flow sheet (e.g. Change in AR.) After each such set of formulas, check to make sure that the audits at the bottom of both sheets still equal zero. Or, write all of the sheet formulas, then all of the cash flow formulas. Either way, **make sure that every balance sheet account is accounted for somewhere in the Cash Flow statement.**
- If multiple BS/CFs are required, write all of the plan version formulas first in one entity. When complete, view the Balance Sheet by plan, and copy the formulas in each account to the other entities. Repeat for the Cash Flow statement.
- When the plan BS/CF is complete, copy the CF formulas into the Actuals version. Some modification to formulas may be necessary, as described below.

Hints

Following are some things to consider when writing formulas in both BS and CF accounts.

Rollup = Yes

Many balance sheet formulas will gather data from the parent plan and all of its subplans, e.g., Accounts Receivable is driven by total company sales. Be sure to use [rollup=yes] in these balance sheet formulas to pull data from the entire Corporate Plan or a parent plan rollup. Otherwise, the formula will pull the data from just the Corporate Plan (Only) or the parent plan (only). Use the Formula Assistant or type "[rollup=yes]" into the formulas.


The corresponding Cash Flow formulas (e.g. Change in AR) do not need to include "[rollup=yes]", because they refer to balance sheet accounts at the same plan level (and not below.) Exceptions include the Net Income row on the Cash Flow sheet. This formula should include "[rollup=yes]" (more on this, below.)

Adaptive Planning, Inc. Balance Sheet/Cash Flow Guide

Multiple Entities

If multiple BS/CFs are required, write all of the plan version formulas first in one entity. When complete, view the Balance Sheet by plan, and paste the formulas in each account to the other entities. Repeat for the Cash Flow statement.

For example, a company with this plan tree wants a BS/CF at five levels: Worldwide G&A, Worldwide R&D, US, UK, and Canada. The BS/CF formulas in these five entities will roll up to the total company BS/CF.



Create the BS/CF formulas first in one entity, e.g. US. When complete, copy them to the other entities.

However, formulas should also be written at the **Corporate Plan (Only)** level, in case P&L data is planned at the Corporate Plan (Only.) For example, some corporate expenses may be planned in the Corporate Plan (Only.) These expenses need to flow through YTD Earnings on the BS/CF for Corporate Plan (Only.)


→ **Note:** Formulas written at the Corporate Plan (Only) level should **NOT** include “[rollup=yes]”. Otherwise, data such as sales driving AR would be double-counted (it would be included in both the Corporate Plan rollup, plus in the individual subplan rollups).

BS Accounts with No Formulas

Balance sheet formulas will be written in a relatively small number of key accounts (e.g. Cash, AR, Fixed Assets, AP, YTD Earnings.)

There are likely to be multiple balance sheet accounts which do not have formulas, but instead will simply carry forward actual balances.

These accounts should be **planned by monthly delta** (which is the balance sheet account default), so they will have this formula:



These accounts will then carry forward an ending balance from the last month of actual data.

Adaptive Planning, Inc. Balance Sheet/Cash Flow Guide

BS Drivers

Custom accounts and/or assumptions can be used to hold entered data (e.g. DSO), or to aggregate other accounts necessary in balance sheet formula-writing. They can also be used as “stepping-stone” calculations, which will become part of a larger calculation. Rather than build a very long formula on the balance sheet, some pieces of the calculation can be created in custom accounts or assumptions. For example, if Average Daily Sales is used to drive Accounts Receivable, a “stepping-stone” custom account could contain this formula to calculate Daily Sales:

=Div(ACCT.Total_Revenue, ASSUM.DaysInMonth)

This Daily Sales account can then be multiplied by a DSO assumption to drive the AR balance.

These custom accounts and assumptions can be gathered in a group called “Balance Sheet Drivers” or “Balance Sheet Assumptions.” They can all be displayed at the bottom of the Balance Sheet.

Custom Account or Global Assumption?

If an entered value (such as DSO) needs to vary for multiple-entity balance sheet planning, be sure to use a custom account, not a global assumption. If only one value is needed for the entire company, a global assumption could be used. However, one set of custom accounts can be used for all drivers. Company-wide numbers can be entered in custom accounts at the Corporate Plan (only) level, like the balance sheet formulas.

Adaptive Planning, Inc. Balance Sheet/Cash Flow Guide

YTD Earnings and Retained Earnings

The Equity section of the balance sheet contains an account called **YTD Earnings (Loss)**. This equity account (or another such account) should have a formula that pulls Net Income from the P&L. There should be a metric account used to calculate Net Income; reference this metric in this formula.

Set this account to be **Planned by Balance**, and write a formula like this:

```
=if(fiscalmonth(this)=1,ACCT.PL_Net_Income[rollup=yes],  
ACCT.YTD_Earnings_Loss[time=this-1]+ACCT.PL_Net_Income[rollup=yes])
```

Note that this formula causes the account to begin accumulating in the first month of the fiscal year, then start over in the first month of each subsequent fiscal year.

Next, write a formula in a **Retained Earnings** account, pulling the previous month's YTD Earnings balance in the first month of each fiscal year. This account should be **Planned by Monthly Delta**, and the formula can look like this:

```
=if(fiscalmonth(this)=1,ACCT.YTD_Earnings_Loss[time=this-1],0)
```

Note that this formula does not need to include [rollup=yes], because it is referencing a balance sheet account at this plan level only, not a P&L account that rolls up data from subplans.

Now go to the **Cash Flow** sheet, and write a formula in the **Net Income (Loss)** account, like this:

```
=ACCT.PL_Net_Income[rollup=yes]
```

Adaptive Planning, Inc. Balance Sheet/Cash Flow Guide

Balance the Balance Sheet

One or more accounts can be formulated to balance the balance sheet (the “plug”). The idea is that this account’s balance should equal the net amount of all the other balance sheet accounts’ balances. This way, total assets will equal total liabilities and equity.

One Account

One account can be used to balance the balance sheet. Typically this is a cash account, but alternatively could be a liability, for example.

Choose the account to use, and set it to be **planned by balance**.


It should contain a formula such as:

$$= \text{ACCT.Liabilities} + \text{ACCT.Equity} - \text{ACCT.Other_Assets}$$

If all balance sheet accounts are being used for planning, there may be multiple cash accounts. Choose the main cash account to hold the formula (this is usually the account with the largest ending balance.) The other cash accounts can be **planned by monthly delta**; these can simply carry forward an ending balance from the last month of actual data.

→ **Hint:** Modify the chart of accounts to gather all other current asset accounts in a rollup called “Current Assets Other Than Cash.” This will simplify the “plug” formula in the main cash account, e.g.

$$= \text{ACCT.Liabilities_Equities} - \\ (\text{ACCT.Curr_Assets_Other_Than_Cash} - \text{ACCT.Fixed_Assets} - \\ \text{ACCT.LongTerm_Assets} - \text{ACCT.Other_Assets})$$




Just make sure that every account on the balance sheet, other than the “plug”, is included in the formula of the “plug.”

Now check the audit metric at the bottom of the Balance Sheet; its value should be zero.

Adaptive Planning, Inc. Balance Sheet/Cash Flow Guide

Next, go to the **Cash Flow Statement** and put formulas in the summary section at the bottom of the sheet:



The **Beginning Cash** row should reference cash on the balance sheet from the prior month, with a formula like this:

=ACCT.1100_Cash[time=THIS-1]

The **Change in Cash** row should reference the total cash flow from all sources, to be calculated above on this cash flow sheet, with a formula like this:

=ACCT.CF_From_All_Sources

Ending Cash should add the total change in cash to Beginning Cash:

=ACCT.CF_Beg_Bal+ACCT.CF_Chg

In the last section, this calculated ending cash is compared to ending cash on the balance sheet. If they are equal, then the cash flow statement correctly picks up all changes on the P&L and Balance Sheet.

The **Cash Balance on Balance Sheet** row should reference cash on the balance sheet in the current month, with a formula like this:

=ACCT.1100_Cash

The **Cash Check** row should have a formula like this:

=ACCT.CF_End_Bal-ACCT.CF_Cash_from_BS

This row should have a value of zero.

➔ **Note:** If an account other than cash, e.g. a liability account, is used to balance the balance sheet, refer to that account in these formulas instead of Cash. These accounts can also be renamed, if desired, to something like Beginning Loan, Change in Loan, Ending Loan, etc.

Adaptive Planning, Inc. Balance Sheet/Cash Flow Guide

Two Accounts

Sometimes two accounts are used to balance the balance sheet, for example, cash and a line of credit (LOC). In cases like this, it is common for the cash balance to be subject to rules, such as “Cash should never be less than zero.” In a case like this, cash is first formulated to follow this rule, and then the liability account is formulated to balance the balance sheet.

To model this, first set the cash account to be **planned by balance**.

Next, in cases like this, it can be helpful to use custom accounts to hold different pieces of the ultimate formula, for audit trail and ease of troubleshooting. Create a group of custom accounts called **Balance Sheet Drivers**, and put this group at the bottom of the balance sheet.

Create a custom account called **Balances Excl Cash**, with a formula like this:

$$\begin{aligned} &=ACCT.Liabilities_Equities- \\ &(\mathbf{ACCT.Curr_Assets_Other_Than_Cash}+ACCT.Fixed_Assets+ \\ &ACCT.LongTerm_Assets+ACCT.Other_Assets) \end{aligned}$$

Note that this is the same formula used above to balance the balance sheet in one cash account.

In the **Cash** account, write a formula such as this:

$$=if(ACCT.BS_Bal_Excl_Cash<0,0,ACCT.BS_Bal_Excl_Cash)$$

In other words, if this calculated “plug” is less than zero, then zero, otherwise the calculated “plug.”

Now, if the calculated plug is a positive amount, the balance sheet balances itself in cash. If the calculated plug is a negative amount, cash is zero and the balance sheet is not balanced.

In the Balance Sheet Drivers custom account group, create a new account called something like **Balances Excl Cash and LOC**. In that account, write a formula like this:

$$\begin{aligned} &(\mathbf{ACCT.Curr_Assets_Other_Than_Cash}+ACCT.Fixed_Assets+ \\ &ACCT.LongTerm_Assets+ACCT.Other_Assets)- \\ &=(\mathbf{ACCT.Curr_Liabs_Excl_LOC}+ACCT.Long_Term_Liabilities+ACCT.Equities)- \end{aligned}$$

Note that some rearranging of the GL Liability account tree may be helpful to isolate the line of credit liability from other current assets, to facilitate this formula.

Next, in the liability account called **Line of Credit**, which should be **planned by balance**, write a formula like this:

$$=if(ACCT.BS_Bal_Excl_Cash_LOC<0,ACCT.BS_Bal_Excl_Cash_LOC,0)$$

In other words, if this calculated “plug” is less than zero, then this calculated plug, otherwise zero.

Now check the BS Check row at the bottom of the Balance Sheet; its calculated value should be zero. It can also be helpful to enter some large positive and negative values into a random account on the balance sheet, to force this calculated plug to be positive or negative, to see if Cash and the Line of Credit behave as expected.

On the **Cash Flow statement**, the section at the bottom should reference both of these accounts.

Adaptive Planning, Inc. Balance Sheet/Cash Flow Guide

Three Accounts

A more complicated scenario can involve using three accounts to balance the balance sheet.

In this example, cash should never be less than zero, but should never be more than a “desired cash” balance such as \$100k. If more cash is available above and beyond the desired balance, the excess cash should be distributed to a parent company. If calculated cash is negative, then cash is zero, and the difference is driven into an intercompany payable account. This intercompany payable should be “paid back,” or reduced, as soon as cash becomes available in subsequent months.

In a scenario like this, several custom accounts are used as stepping stones. In the Balance Sheet Drivers group of custom accounts, create an account called something like **ENTER Desired Cash Balance**. This is where the cash “ceiling” of \$100k is entered.

Next, create a custom account called **Balances Excl Cash, IC Payable, Dist**. Write a formula that picks up all balances on the balance sheet except for the three accounts being used in this three-account-plug. Create another account called **Balances Excl Cash and IC Payable plus Desired Cash**. Write a formula that picks up all balances on the balance sheet except for Cash and the intercompany payable account, plus the desired cash balance. Create a third account called **Balances Excl Cash and Interco Payable**. Write a formula that picks up all balances on the balance sheet except for Cash and the intercompany payable account.

Set the **Cash** account to be planned by balance, and write the following formula:

```
=if((ACCT.BS_Bal_Excl_Cash__IC_Pay_Dist)<0,0,  
    if((ACCT.BS_Bal_Excl_Cash__IC_Pay_Dist)>0,  
    if((ACCT.BS_Bal_Excl_Cash__IC_Pay_Dist)<ACCT.BS_Desired_Cash_Bal,  
    (ACCT.BS_Bal_Excl_Cash__IC_Pay_Dist), ACCT.BS_Desired_Cash_Bal),  
    (ACCT.BS_Bal_Excl_Cash__IC_Pay_Dist)))
```

In other words, if the “plug” is less than zero, then cash should be zero. Otherwise, if the plug is greater than zero, then if the plug is less than the desired cash balance, then the plug, otherwise, the desired cash balance. Otherwise, the plug (in the situation where the plug is greater than zero but less than desired cash.)

At this point it can be helpful to enter some large positive and negative values into a random account on the balance sheet, to force this calculated plug to be positive or negative, to see if Cash behaves as expected.

Now write a formula like this in the equity **Distribution** account, which should be **planned by monthly delta** (previous plus):

```
=if(ACCT.BS_Bal_Excl_Cash__IC_Pay_Dist_plus_Des_Cash>0,0,  
    ACCT.BS_Bal_Excl_Cash__IC_Pay_Dist_plus_Des_Cash-  
    ACCT.Distributions[time=THIS-1])
```

In other words, if less cash is available than the desired balance, then zero, otherwise, if more cash is available than the desired balance, this account should increase by the available cash in the current month.

Adaptive Planning, Inc. Balance Sheet/Cash Flow Guide

Finally, in the **Intercompany Payable** account, which should be **planned by balance**, write a formula like this:

```
=if(ACCT.BS_Bal_Excl_Cash_and_Interco_AP>0,ACCT.BS_Bal_Excl_Cash_and_Interco_AP,0)
```

In other words, if the calculated “plug” is greater than zero, then the calculated plug, otherwise zero. This is the final plug on the balance sheet, which should now balance. Again, it can be helpful to enter some large positive and negative values into a random account on the balance sheet, to force this calculated plug to be positive or negative, to see if all three accounts (Cash, Distributions, and Intercompany Payable) behave as expected under different scenarios.

On the **Cash Flow statement**, the section at the bottom should reference all three of these accounts.

A Different Approach

Rather than use one or more accounts to balance the balance sheet, as described above, a different approach is to have balance sheet cash equal cash flow ending cash. In this case, write the cash flow formulas as described in this document. The ending cash on the cash flow statement equals the change in all P&L and balance sheet (other than cash) activity. Write a formula in the balance sheet cash account to reference calculated cash on the Cash Flow statement.

This approach can also be used if cash should not drop below zero (use a Line of Credit to prevent this.) If ending cash from the cash flow statement is less than zero, then send this amount to the Line of Credit.

Adaptive Planning, Inc. Balance Sheet/Cash Flow Guide

Accounts Receivable

Following are two examples of commonly used methodologies for planning AR.

Method #1: AR is planned as DSO x average daily sales. In this case, AR is **planned by balance**.

First, create a custom account (or assumption) in the Balance Sheet Drivers account group, to hold the **number of business days** in each month. Then, in a second custom account or assumption, enter **DSO**. Then, in a third account, calculate **Average Daily Sales** with a formula such as:

$$=Div(ACCT.Total_Revenue,ACCT.DaysInMonth)$$

If revenue swings from month to month, a rolling two month average could be used, for example, with a formula such as this:

$$=Div((ACCT.Total_Revenue[rollup=yes]+ ACCT.Total_Revenue[time=this-1, rollup=yes]),
(ACCT.DaysInMonth+ ACCT.DaysInMonth[time=this-1]))$$

In other words, take this months' and last months' revenues, divided by this months' and last months' number of days.

→ **Note:** In this case, be sure to enter the number of business days and the Average Daily Sales formulas in the Actuals version as well, because the formulas in the first plan month will be referring to a prior month in the Actuals version.

Then, on the balance sheet, in the Accounts Receivable account, the formula is:

$$=ACCT.DSO*ACCT.Avg_Daily_Sales$$


→ **Hint:** The DSO account can be calculated in the Actuals version, for reference. If this is desired, go into the Actuals version and enter the number of business days, and copy the formula from the plan version for Average Daily Sales. Then write a formula like this in the DSO account:

$$=div(ACCT.Accounts_Receivable, ACCT.Avg_Daily_Sales)$$

Adaptive Planning, Inc. Balance Sheet/Cash Flow Guide

Method #2: In this case, AR is **planned by monthly delta**. Its increase is driven from modeled invoicing from a Sales sheet(s). Its decrease is also driven from invoicing, but from a time period other than the current month.

When writing the decrease portion of the formula, to factor in the average number of days that invoices are outstanding, use the Month option in the Formula Assistant.



For example, if Days Sales Outstanding (DSO) is 20 days, use time=THIS-1 (which refers to 30 days ago, so this is approximate.) If DSO is 60 days, use time=THIS-2, etc. For example:

=ACCT.Total.Invoicing[rollup=yes]-ACCT.Total.Invoicing[time=THIS-1,rollup=yes]

→ **Hint:** The “decrease” part of the formula should be modified for the first month or two of the plan period. E.g. if AR is decreased by ACCT.Total.Invoicing[time=THIS-2,rollup=yes], the formula would be looking for modeled invoicing from two months ago. No modeled invoicing exists in the months prior to the beginning of the planning year. So, in the first two months of the plan year, this part of the formula should be replaced by a value.

Allowance for Doubtful AR

Allowance for doubtful AR accounts is typically **planned by balance**, with a formula such as:

=ACCT.AR*(-1*ACCT.Doubt_AR_Pct)

The value in the custom account (or assumption) called Doubt_AR_Pct can be entered or calculated. It is the percentage of AR which is to be considered doubtful, or uncollectable.

If/when bad debts are written off, entries can be made in the account to decrease the balance.

Hint: The change in this balance sheet account may be pulled into the Bad Debt expense account in the P&L. Or, conversely, the Bad Debt Expense can be planned first. Then the Allowance for Doubtful AR can be **planned by monthly delta**, with a formula increasing it each month by the amount of the expense.

Adaptive Planning, Inc.

Balance Sheet/Cash Flow Guide

Prepaid Expenses

A company with significant prepaid expenses may want to plan in detail the timing of the payments, and the corresponding amortization of the payments into expense over the life of the prepaid asset. This section describes how to approach this.

Use a Modeled Sheet

Prepaid expenses can be planned using a modeled sheet, which is similar to a Capital sheet. Users will enter the payments on the sheet, and these values will be spread into expense over the life of the prepaid asset (typically 6 months or a year.) On the balance sheet, the prepaid asset accounts will be increased by the payments, and decreased by the associated expense.

First, **consider the types of prepaids and their lives**, or how the prepaid expenses will be amortized to the P&L. The most common life is 12 months, such as in the case of an annual insurance premium. Create a dimension with all of the possibilities, similar to the Asset Class dimension used on the Capital sheet. Call this dimension something like Prepaid Class. Put the dimension on the modeled sheet and create a spread lookup table associated with it.

Note: If different types of prepaids are going to be amortized into different expense accounts, create a dimension value for each type, even if they have the same life, e.g. Insurance 12 months and Software Maintenance 12 months.

Create a timeline on the sheet. This will be used to hold the payments.

Create a modeled account called Prepaid Amortization, and spread the values entered in the timeline on the sheet using the spread lookup table.

Next, **review the accounts which are to be used for the expense amortization**, e.g. insurance expense. As with depreciation, these accounts will be planned in two separate ways. Amortization of existing prepaids will be entered or imported. Amortization of future (planned) prepaids will be automatically calculated by the Prepaid sheet.

Go to the expense accounts which are to receive the amortization of prepaids. Create two sub-accounts under the existing expense account. Name one something like “Amort Old Prepaids” and the other something like “Amort New Prepaids.”

→ **Hint:** Name the first new sub-account something like “Amort Old Prepaids.” This is because any values that reside in a non-rollup account will be moved to the first new sub-account created. Therefore, if there are any actual or old plan values in the original expense account, they will be moved to the first new sub-account, “Amort Old Prepaids.”

→ **Hint:** A third sub-account may also need to be created, if the account is also used to plan expenses unrelated to Prepaids. This is sometimes the case with Insurance expense accounts. For example:

73020_00 Ins - Auto	73020_00	Expense
73020_01 Auto Exp from Old Ppds	73020_01	Expense
73020_02 Auto Exp from New Ppds	73020_02	Expense
73020_03 Auto Exp not from Ppds	73020_03	Expense

Adaptive Planning, Inc. Balance Sheet/Cash Flow Guide

Next, **link the modeled amortization account to the appropriate GL accounts** (the ones that have "New Prepays" in their names.) In the Formulas tab, go to the "New Prepays" GL accounts and write formulas equal to the modeled amortization accounts, such as:

=ACCT.Prepaid_Amort[Prepaid_Class=Auto]

Next, **enter amortization of existing prepaids** into the expense accounts which have "Old Prepays" in their names. Note that the sum of this expense should equal the last actual value in the prepaid asset account(s) on the balance sheets.

Plan the Prepaid Balance Sheet Accounts

Prepaid expenses are **planned by balance**. For each prepaid, create three splits or sub-accounts, with names and formulas such as:

1400 Prepaid Expenses			
1401 Prepaid Insurance	3,007		
Beginning Balance	0	3,007	-150
Plus Additions	0	0	0
Less Expensed	0	-3,157	-657
Total	3,007	-150	-808

Beginning Balance: Refers to this same account balance, last month:

=ACCT.Prepaid_Ins_Auto[time=THIS-1]

Plus Additions: Refers to the modeled account which captures the value of new prepaid expenses (the payments) from the Prepaid sheet:

=ACCT.Ins_Auto.Pmts[rollup=yes]

Less Decreases: Refers to the total expense related to both old and new prepaids (notice that this excludes the possible third sub-account, which is used for the same type of expense unrelated to prepaids):

=-1*(ACCT.Auto_Ins_Exp_Old_Ppds[rollup=yes]
+ACCT.Auto_Ins_Exp_New_Ppds[rollup=yes])

Create a Prepaid Audit Sheet

Especially if there are multiple prepaid balance sheet accounts, and multiple classes of prepaids being amortized to many different expense accounts, it can be helpful to create a standard sheet whose purpose is to audit all of the prepaid activity. This sheet can be called something like "Audit Prepays." Following is a suggested layout of accounts to include on the report:

1. **Amortization of Old Prepays:** Include the GL expense accounts which contain amortization expense from prepaids which existed prior to the plan period.
2. **Amortization of New Prepays:** Include the GL expense accounts which contain amortization expense from new prepaids, planned on the Prepaid (Capital) sheet. The sum of #1 and #2 should equal the decreases from the prepaid balance sheet accounts; in other words, does the balance sheet decrease properly reflect the expense in the P&L?

Adaptive Planning, Inc. Balance Sheet/Cash Flow Guide

3. **Amortization of New Prepaids (Modeled):** Include the modeled amortization account from the Prepaid sheet. The total of #2 should equal the total of #3; in other words, does the GL expense (and consequently, balance sheet) activity properly reflect the values calculated by the modeled sheet?
4. **Additions to Prepaids (Modeled):** Include the modeled “value” accounts from the Prepaid sheet. This should equal the additions to the prepaid balance sheet accounts; in other words, does the balance sheet properly reflect the increases in prepaid expenses?
5. **Prepaids (Balance Sheet):** Include the balance sheet prepaid accounts. Use this section to check the balance sheet activity against the accounts listed above.

Fixed Assets

Fixed assets, also known as Property, Plant, and Equipment, are **planned by Monthly Delta**. These accounts are increased by the additions to the Capital sheet. There are no decreases to these accounts (because the decreases are planned in the offsetting Accumulated Depreciation accounts.)

If there is a separate asset account for each asset class on the Capital sheet, then the formula in each asset account should look like this:

=ACCT.Capex[Asset_Class=Off_FFE,rollup=yes]

This formula pulls capital spending from the Capital sheet for one asset class; in this case, Office Furniture & Equipment.

If there is one asset account for all fixed assets, then the formula in this asset account should look like this:

=ACCT.Capex[rollup=yes]

This formula pulls the modeled value of all fixed assets from the Capital sheet, for all asset classes.

Accumulated Depreciation

Accumulated Depreciation is **planned by Monthly Delta**. This account has a negative balance and is increased by a negative number, with a formula such as:

=-1*ACCT.Total_Depreciation[rollup=yes]

This formula pulls the amounts from all GL depreciation accounts. Please see the section of the Implementer’s Guide called “Capital Planning and Depreciation Expense,” for ideas on how to create Depreciation Expense sub-accounts for existing assets and new assets. The Depreciation expense rollup account hereby includes both depreciation on **existing assets** and depreciation on **new assets** planned on the Capital sheet.

→ **Hint:** Don’t forget to make the results of this formula a negative number.

→ **Hint:** Even if new capital spending is planned by asset class, often the depreciation on existing assets is brought in by plan, but not by asset class. In this case, total depreciation expense cannot be broken out by asset class. Depreciation expense can be pulled into one accumulated depreciation account for all asset classes. If accumulated depreciation needs to be broken out by specific asset classes, **then depreciation on existing assets must be entered by asset class**. In this case, use splits in the depreciation accounts, and tag the splits with the Asset Class dimension. Then the formulas in the accumulated depreciation accounts can look like this:

Adaptive Planning, Inc. Balance Sheet/Cash Flow Guide

$$\begin{aligned} &= -1 * (\text{ACCT.Deprec_Exist_Asset}[\text{Asset_Class}=\text{FFE}, \text{rollup}=\text{yes}]) \\ &+ \text{ACCT.Deprec_New_Asset}[\text{Asset_Class}=\text{FFE}, \text{rollup}=\text{yes}] \end{aligned}$$

Where ACCT.Deprec_Exist_Asset is the GL sub-account, and ACCT.Deprec_New_Asset is the modeled account from the Capital sheet.

Accounts Payable

Following are two examples of commonly used methodologies for planning AP.

Method #1: AP is planned as ADP x average daily spending. In this case, AP is **planned by balance**.

First, create a custom account (or assumption) in the Balance Sheet Drivers account group, to hold the **number of business days** in each month. (This account may have already been setup as part of Accounts Receivable planning.) Then, in a second custom account or assumption, enter **ADP**. Then, in a third account, calculate **Expenses that Run through AP** with a formula such as:

$$\begin{aligned} &= (\text{ACCT.Cost_Of_Goods_Sold}[\text{rollup}=\text{yes}] - \\ &\quad (\text{ACCT.Expenses}[\text{rollup}=\text{yes}] - \text{ACCT.5000_Salaries}[\text{rollup}=\text{yes}] - \\ &\quad \text{ACCT.5160_Payroll_Tax_Expense}[\text{rollup}=\text{yes}] - \text{ACCT.5065_Comms}[\text{rollup}=\text{yes}] - \\ &\quad \text{ACCT.5110_DeprecAmort}[\text{rollup}=\text{yes}] - \text{ACCT.5250_Bad_Debt_Expense}[\text{rollup}=\text{yes}] - \\ &\quad \text{ACCT.Allocations}[\text{rollup}=\text{yes}]) \end{aligned}$$

In other words, summarize the accounts that are paid through Accounts Payable; these typically exclude salary related accounts and non-cash accounts such as depreciation and bad debt expense.

→ **Hint:** Modify the chart of accounts as needed to group expenses to facilitate gathering accounts which drive AP.

Then, on the balance sheet, in the Accounts Payable account, the formula is:

$$= \text{ACCT.ADP} * \text{ACCT.Exp_that_run_through_AP}[\text{rollup}=\text{yes}]$$

→ **Hint:** The ADP account can be calculated in the Actuals version, for reference. If this is desired, go into the Actuals version and enter the number of business days, and copy the formula from the plan version for Expenses that Run through AP. Then write a formula like this in the ADP account:

$$= \text{div}(\text{ACCT.Accounts_Payable}, \text{ACCT.Exp_that_run_through_AP}[\text{rollup}=\text{yes}])$$

Method #2: AP is **planned by Balance**. It is calculated as spending that runs through AP (as described above), multiplied by a factor.

To factor in the average number of days that invoices are outstanding, multiply the entire formula by a percentage. For example, expenses * 1 means that typically one month of expenses is unpaid. Expenses * .50 means that payables are paid fairly quickly, because only ½ of one month is unpaid. Expenses * 1.10 means that payables take slightly longer than a month to pay. For example:

$$= (\text{ACCT.Expenses}[\text{rollup}=\text{yes}] - \text{ACCT.Salary.Related}[\text{rollup}=\text{yes}])$$

Adaptive Planning, Inc. Balance Sheet/Cash Flow Guide

-ACCT.DeprecExp[rollup=yes])*1.10

Alternatively, the formula can reference the appropriate expenses from the current month plus some portion of the same expenses from prior month(s), such as:

=ACCT. Exp_that_run_through_AP [rollup=yes]+
ACCT.AP_Exp_Driver[time=THIS-1,rollup=yes]*.5


In this example, the AP balance is planned as these expenses for the current month plus ½ of these expenses from the prior month. In other words, payables are normally paid in 45 days. (If this type of formula is used, be sure to modify it in the first month of the planning year: replace the second piece of the formula with a value, since there are no planned expenses prior to this month.)

→ **Note:** In this case, be sure to enter the Expenses that Run through AP formulas in the Actuals version as well, because the formulas in the first plan month will be referring to a prior month in the Actuals version.

Other Payables

There may be other payables which need to be formulated. An example is Bonus Payable. If bonus expense is planned in the P&L, this expense will drive the payable.

Bonus Payable is **planned by monthly delta**. It is increased each month by the amount of the bonus expense. Its formula would look like this:



Previous + =ACCT.Ann_Bonus_Exp[rollup=yes]

Ascertain when bonuses are paid to employees. In this example, they are paid in the first month of each quarter. The payout is equal to the accrued expense from the previous quarter. So the formula in the first month of each quarter would look like this:

=ACCT.Qtrly_Bonus[rollup=yes]-(ACCT.Qtrly_Bonus[time=**THIS-1**,rollup=yes]+
ACCT.Qtrly_Bonus[time=**THIS-2**,rollup=yes]+ACCT.Qtrly_Bonus[time=**THIS-3**,rollup=yes])

Alternatively, the decrease in the account can reference the balance in the bonus payable account from the month before:

=ACCT.Qtrly_Bonus[rollup=yes]-(ACCT. Bonus_Payable[time=**THIS-1**]

→ **Hint:** The “decrease” part of the formula should be modified for the first month of the first quarter of the plan year. No modeled bonus expense exists in the months prior to the beginning of the planning year. So this part of the formula should be replaced by a value.

→ **Hint:** Bonus Payout may have already been calculated in the Personnel sheet, since Bonus Payout is what affects Payroll Tax calculations. If this is the case, reference the Bonus Payout in the decrease portion of the Bonus Payable account.

Adaptive Planning, Inc. Balance Sheet/Cash Flow Guide

Deferred Revenue

Companies that recognize revenue over time, e.g. software companies, may carry a Deferred Revenue asset liability on their balance sheet. Deferred revenue is increased when a contract is booked, and then decreased over time as the revenue is recognized. Following is an example of how deferred revenue can be planned. This example assumes that revenue is planned on a modeled sales sheet, where invoicing and recognized revenue are calculated with spread lookup tables.

Deferred Revenue is **planned by balance**. Its balance can be calculated as the difference between what has been invoiced to-date and what has been recognized as revenue to-date.

This calculation can be accomplished in custom accounts. **First**, create a cumulative account linked to total modeled revenue. The balance in this account increases each month by the amount of modeled revenue. **Second**, create another cumulative account linked to total modeled invoicing. The balance in this account increases each month by the amount of modeled invoicing. Create a **third** periodic account, whose formula subtracts cumulative revenue from cumulative invoicing. This last account calculates the deferred revenue balance. This can be pulled onto the balance sheet.

Accounts	Jan-2009	Feb-2009	Mar-2009
Deferred Revenue			
Cumulative Revenue	5,550,125	6,166,833	6,783,542
Cumulative Invoicing	7,571,250	7,601,250	7,631,250
Deferred Revenue Balance	2,021,125	1,434,417	847,708

→ **Note:** The beginning balance in Deferred Revenue, carried forward from the last month of actual data, needs to be amortized into revenue. A sub-account dedicated to this purpose can be created on the balance sheet. The revenue amortization can be entered here, then pulled into the P&L, or it can be entered on the P&L, then pulled onto the balance sheet. Either way, consider the amortization of the beginning balance and make sure its decrease is integrated appropriately with the P&L.

Adaptive Planning, Inc. Balance Sheet/Cash Flow Guide

Deferred Income Tax

There may be a balance sheet account which should be an asset if it has a debit balance, or a liability if it has a credit balance. An example is Deferred Income Tax.

First, make sure there is an asset account (e.g. 25000_00 Deferred Tax Benefit) and a corresponding liability account (e.g. 29000_00 Income Taxes Payable.) The monthly balance will be displayed in only one of these two accounts, depending on whether it is a debit or a credit.

In this example, the amount of the deferred tax is first entered or calculated in the asset account. It does not matter whether this account is planned by monthly delta, or by balance.

Under the asset account, create a new sub-account called something like 25000_adj Display Only if Debit Balance. This account is **planned by balance**.

Deferred Tax Benefit	
25000_00 Federal Income Taxes - Current	
25000_adj Display Only if Debit Balance	
Total	

On the balance sheet, create a formula in this new account such as:

=IF(ACCT.25000_00<0,-1*ACCT.25000_00,0)

In other words, if the balance in the Deferred Tax account is less than zero (this would be a credit balance, since this account is an asset), then this adjustment equals that balance x -1. In this case, the adjustment will zero out the parent account, as shown here:

Deferred Tax Benefit	
25000_00 Federal Income Taxes - Current	-45,000
25000_adj Display Only if Debit Balance	45,000
Total	0

But if the balance is greater than zero (a debit balance), then the adjustment is zero. In this case, the parent account equals the balance in the Deferred Tax account, as shown here:

Deferred Tax Benefit	
25000_00 Federal Income Taxes - Current	45,000
25000_adj Display Only if Debit Balance	0
Total	45,000

Then, create a formula like this in the liability account (which is **planned by balance**):

=-ACCT.25000_adj

This will cause the liability to have a credit balance in the amount of the adjustment out of the asset.

The result is if the Deferred Tax asset account has a credit balance, the asset account is zeroed out, and the liability assumes the balance as a credit.

Adaptive Planning, Inc. Balance Sheet/Cash Flow Guide

Other Deferrals

There may be other deferrals of revenue and/or cost on the balance sheet. These can be calculated using formulas similar to those in the Deferred Revenue account.


An example is a company which plans construction job revenue and related billings. As the job progresses, the customer is billed several times, and revenue is recognized along the way, depending on how much of the job is completed. At any given point during the job, total revenue recognized to date may exceed what the customer has been billed, or vice versa.

If accumulated revenue exceeds accumulated billing, the difference is an asset, because this is a receivable which hasn't been booked yet. If accumulated billing exceeds accumulated revenue, the difference is a liability, because a receivable has been booked but the revenue has not been recognized.

The asset and the liability can be planned by using custom accounts behind the scenes to accumulate revenue and billings, calculate the difference, and use IF statements to determine whether the calculated value should be an asset or a liability.

How to Accumulate a Revenue or Expense Account

Revenue and expense accounts are periodic accounts; that is, the accounts display the net activity in each month. What is needed here is an account which displays the accumulated (to-date) value in each month. To accomplish this, create a custom account called something like "Accumulated Revenue." In account details, select type of Cumulative.



Using the Formulas tab, create a formula in the account (in the appropriate departments) which references the appropriate modeled or GL revenue accounts, such as:

=ACCT.Total_Revenue

→ **Note:** This accumulation will span more than one plan year, as balance sheet accounts do. This is appropriate for this particular scenario, because a job can begin in one year and continue into the next.

Create a similar account and formula to accumulate total billings.

Next, create a custom account called something like "Accum Revenue in Excess of Billings." Create a formula in the account (in the appropriate departments) such as:

=IF(ACCT.Accum_Rev-ACCT.Accum_Bill>0,ACCT.Accum_Rev-ACCT.Accum_Bill,0)

In other words, if accumulated revenue is greater than accumulated billings, display this difference. Otherwise, display zero. This calculated value will be the balance in the **asset** account.

Adaptive Planning, Inc. Balance Sheet/Cash Flow Guide

Create a similar custom account called something like "Accum Billing in Excess of Revenue."
The formula should be like this:

=IF(ACCT.Accum_Rev-ACCT.Accum_Bill<0,-1*(ACCT.Accum_Rev-ACCT.Accum_Bill),0)

In other words, if accumulated billings are greater than accumulated revenue, display this difference. Otherwise, display zero. This calculated value will be the balance in the **liability** account.

The last step is to go to the balance sheet accounts, and link them to these two calculations. These accounts should be **planned by balance**. The asset account should contain a formula such as this:

=ACCT.Job_Accum_Rev_Excess_Bill[rollup=yes]

The liability account should contain a formula such as this:

=ACCT.Job_Accum_Bill_Excess_Rev[rollup=yes]

Adaptive Planning, Inc. Balance Sheet/Cash Flow Guide

6. Balance Sheet Issues with Multiple Currency

In an Adaptive Planning model using multiple currencies, all GL accounts automatically convert from a subplan's local currency to the currency of the parent department, and ultimately to the currency of the total company. The exchange rates used for this conversion are assigned to plans, and are managed by the Administrator. (For more information on this topic, please see the User Guide for International Users and Planning in Multiple Currencies.)

There may be times when this automatic assignment of exchange rates is not appropriate in Balance Sheet planning. Instead, such as in the case of Common Stock, YTD Net Income, and Retained Earnings, it can be more appropriate for an historical exchange rate to be used convert values from local to corporate currency.

Example #1

For example, consider a company with this sample plan rollup, with balance sheets planned at each entity:

- US Parent Company (USD)
 - US Operations (USD)
 - European Operations (EURO)
 - Canadian Operations (CAD)

The default behavior of the **YTD Net Income** equity account in Adaptive Planning is this:

January YTD Net Income is converted to USD, using January's exchange rate. February YTD Net Income is converted to USD using February's exchange rate, etc.

But the desired behavior of YTD Net Income may be this:

January YTD Net Income is converted to USD using January's exchange rate. February YTD Net Income should be converted as (January Net Income converted to USD using January's exchange rate) plus (February Net Income converted to USD using February's exchange rate), etc.

Another example is common stock, where the desired behavior is for stock to remain unchanged with currency exchange fluctuations.

Another balance account, called CTA (Cumulative Translation Adjustment or Gain /Loss on FX), can hold the deltas between these calculations and what the YTD Net Income or Common Stock balances would be, using the monthly currency exchange rates.

Please see **Appendix B** for an example of how to calculate Common Stock separate from Cumulative Translation Adjustment. The same calculations can be applied to YTD Net Income and CTA.

Example #2

In the following examples, actual or planned values reside in local currency at the subplan level, e.g. Luxembourg, planned in EUROS. The corporate currency is USD. There is an Elimination plan corresponding to each foreign subplan, e.g. Luxembourg Eliminations, planned in USD.

The purpose of the Elimination plan is to hold formulas that reverse the automatic currency exchange, and instead apply historical exchange rates to certain accounts.

Adaptive Planning, Inc. Balance Sheet/Cash Flow Guide

First create global assumptions to hold the Historical Exchange Rates:

Historical Exchange Rates				
Historical FX Euro Note w Inc	1.470400	1.470400	1.470400	1.470400
Historical FX Euro Goodwill	1.3789517	1.3789517	1.4729000	1.4729000
Historical FX Euro DTA	1.384379	1.384379	1.384379	1.384379
Historical FX Euro Invest in UK	1.470588	1.470588	1.470588	1.470588
Historical FX Euro Invest in APAC	1.297872	1.297872	1.297872	1.297872
Historical FX Euro APIC	1.443911	1.443911	1.493182	1.560900
Historical FX Euro CTA	1.472902	1.472902	1.472902	1.472902
Historical FX Euro Beg 2008 RE	1.427767	1.427767	1.427767	1.427767
Historical FX GBP APIC	1.993939	1.963100	1.995100	1.982000
Historical FX GBP Beg 2008 RE	1.861756	1.861756	1.861756	1.861756
Historical FX SGD Beg RE	0.640364	0.640364	0.640364	0.640364
Historical FX GBP Common Stock	2.000000	2.000000	2.000000	2.000000
Historical FX Ireland NP				
Historical FX Ireland CS				

Then, in the USD Elimination plan, write formulas such as this, in the Long Term Asset account called Goodwill:

$$\begin{aligned}
 &= ((\text{div}(\text{ACCT}.15000_Goodwill[\text{plan}=\text{Balance Sheet-Lux (Euro)}], \\
 &\quad \text{ASSUM.ExchangeRate.EUR.E}) - \\
 &\quad \text{div}(\text{ACCT}.15000_Goodwill[\text{plan}=\text{Balance Sheet-Lux (Euro)}, \text{time}=\text{THIS}-1], \\
 &\quad \text{ASSUM.ExchangeRate.EUR.E}[\text{time}=\text{THIS}-1])) * \text{ASSUM.Hist_FX_Eur_GW} \\
 &\quad + (\text{ACCT}.15000_Goodwill[\text{plan}=\text{Balance Sheet-Lux (USD)}, \text{time}=\text{THIS}-1]) \\
 &\quad - \text{ACCT}.15000_Goodwill[\text{plan}=\text{Balance Sheet-Lux (Euro)}]
 \end{aligned}$$

In other words, first divide Goodwill from LUX (already converted to USD) by the system ending exchange rate. This has the effect of stating Goodwill in Euros. Do the same thing for Goodwill last month. Subtract last month's Goodwill in Euros from this month's Goodwill in Euros to calculate the change in Goodwill in Euros, then multiply this by the historical exchange rate.

Now take last month's Goodwill balance from the rollup plan that sums LUX and LUX Elim, and subtract this month's good will balance from LUX. Note that the values in this part of the formula have been automatically converted to USD using the system exchange rates.

Adaptive Planning, Inc. Balance Sheet/Cash Flow Guide

Explore Cell

Account: Goodwill Code: 15000_Goodwill Type: Long Term Asset Rolls up to: Long Term Assets	Value: <h3 style="margin: 0;">182,487 USD</h3>
Plan: Balance Sheet-Lux (USD Elim) Rolls up to: Balance Sheet-Lux (USD)	Formula: $((\text{div}(\text{ACCT.15000_Goodwill}[\text{plan}=\text{Balance Sheet-Lux (Euro)}], \text{ASSUM.ExchangeRate.EUR.E})-\text{div}(\text{ACCT.15000_Goodwill}[\text{plan}=\text{Balance Sheet-Lux (Euro)}], \text{ASSUM.ExchangeRate.EUR.E})))$
Month: Feb 2008 (Actuals) Rolls up to: Q1-FY08	Visible on the following sheets: Balance Sheet

Formula	
Formula:	$((\text{div}(\text{ACCT.15000_Goodwill}[\text{plan}=\text{Balance Sheet-Lux (Euro)}], \text{ASSUM.ExchangeRate.EUR.E})-\text{div}(\text{ACCT.15000_Goodwill}[\text{plan}=\text{Balance Sheet-Lux (Euro)}], \text{ASSUM.ExchangeRate.EUR.E})))$
Evaluated Formula:	$((\text{div}(-1,364,406, 2)-\text{div}(-1,364,406, 1))*1.3789517)+(-1,881,450)-1,364,406$

This total calculated value is expressed in USD, because the elimination plan is in the USD currency. This calculated value is added to the LUX converted value in Goodwill when the two plans are rolled up. So this entry has the effect of increasing or decreasing the total, rolled up Goodwill balance.

This same treatment may be applied to accounts such as a long-term deferred tax asset, equity investments in other companies, or to retained earnings from prior years.

A liability account such as **Intercompany Notes Payable** might have a subset of the formula described above, such as this:

$$\begin{aligned}
 &= (\text{div}(\text{ACCT.29400_Interco_Notes_Payable}[\text{plan}=\text{Balance Sheet-Lux (Euro)}], \\
 &\quad \text{ASSUM.ExchangeRate.EUR.E}) * \\
 &\quad \text{ASSUM.Hist_FX_Eur_Note_Inc}) \\
 &\quad - \text{ACCT.29400_Interco_Notes_Payable}[\text{plan}=\text{Balance Sheet-Lux (Euro)}]
 \end{aligned}$$

In other words, first divide IC Notes Payable from LUX (already converted to USD) by the system ending exchange rate. This has the effect of stating IC Notes Payable in Euros. Then multiply this by the historical exchange rate. Subtract the prior month's IC Notes Payable from LUX (already converted to USD) by the system ending exchange rate.

Explore Cell

Account: Interco Notes Payable Code: 29400_Interco_Notes_Payable Type: Long Term Asset Rolls up to: Interco Notes Payable	Value: <h3 style="margin: 0;">202,755 USD</h3>
Plan: Balance Sheet-Lux (USD Elim) Rolls up to: Balance Sheet-Lux (USD)	Formula: $(\text{div}(\text{ACCT.29400_Interco_Notes_Payable}[\text{plan}=\text{Balance Sheet-Lux (Euro)}], \text{ASSUM.ExchangeRate.EUR.E}) * \text{ASSUM.Hist_FX_Eur_Note_Inc}) - \text{ACCT.29400_Interco_Notes_Payable}[\text{plan}=\text{Balance Sheet-Lux (Euro)}]$
Month: Feb 2008 (Actuals) Rolls up to: Q1-FY08	Visible on the following sheets: Balance Sheet

Formula	
Formula:	$(\text{div}(\text{ACCT.29400_Interco_Notes_Payable}[\text{plan}=\text{Balance Sheet-Lux (Euro)}], \text{ASSUM.ExchangeRate.EUR.E}) * \text{ASSUM.Hist_FX_Eur_Note_Inc}) - \text{ACCT.29400_Interco_Notes_Payable}[\text{plan}=\text{Balance Sheet-Lux (Euro)}]$
Evaluated Formula:	$(\text{div}(-1,701,254, 2)*1.4704500) - 1,701,254$

Copyright © 2004-2011 Adaptive Planning, Inc. All rights reserved.

Adaptive Planning, Inc. Balance Sheet/Cash Flow Guide

Then the Equity account called Cumulative Translation Adjustment could have a formula such as this:

Equity: Cumulative Translation Adjustment

```
=-ACCT.Investment_in_Subs[plan=Balance Sheet-Lux (USD)]-  
ACCT.30015_Paid_In_Capital[plan=Balance Sheet-Lux (USD)]-  
ACCT.36000_CTA_for_tax[plan=Balance Sheet-Lux (USD)]-  
ACCT.Retained_Earnings[plan=Balance Sheet-Lux (USD)]-  
ACCT.YTD_Earnings_Loss[plan=Balance Sheet-Lux (USD)]+  
ACCT.Assets[plan=Balance Sheet-Lux (USD)]-  
ACCT.Liabilities[plan=Balance Sheet-Lux (USD)]
```

In other words, from the rollup plan that sums LUX and LUX Elim, take the converted balances in USD in all accounts excluding this account.

Adaptive Planning, Inc. Balance Sheet/Cash Flow Guide

7. Cash Flow Statement

The cash flow statement accounts for changes in the balance sheet (which includes net income.) Every account on the balance sheet must be referenced somewhere on the cash flow statement. If anything is left out, the cash flow statement will be out of balance. For more information on this topic, see the section below, called "Troubleshooting the Cash Flow Statement."

This section will illustrate some typical formulas for the cash flow accounts.

Net Income (Loss)

Net Income (CF_NI) should be linked to the same net income formula which is used on the balance sheet, as described above in the section called YTD Earnings and Retained Earnings.

Net Income Adjustments

Depreciation & Amortization (CF_Deprec_Amort) should be linked to total depreciation and amortization expense on the P&L. The formula should look something like this:

=ACCT.Deprec_Amort[rollup=yes]

Do not include expenses amortized from prepaids; the change in prepaids will be addressed separately. Do not include any amortization which is addressed in separate lines on the Cash Flow statement, such as change in long term assets.

Change in Operating Assets & Liabilities

Calculate the change from the prior month for non-cash operating assets and liabilities.

→ **Hint:** For asset accounts, the formula should equal **prior** month's ending balance minus **current** month's ending balance. For liability accounts, the formula should equal **current** month's ending balance minus **prior** month's ending balance (unless liabilities are planned with a negative balance; in which case, the formula is the same as for assets.)

Adaptive Planning, Inc. Balance Sheet/Cash Flow Guide

Following are some examples, which assume that liabilities are planned with a positive balance.

Incr/(Decr) Accounts Receivable (CF_Incr_Decr_AR):

$$=ACCT.AR[time=THIS-1]-ACCT.AR$$

Incr/(Decr) Prepaids (CF_Incr_Decr_PPD):

$$=ACCT.PPDs [time=THIS-1]-ACCT.PPDs$$

Incr/(Decr) Other Assets (CF_Incr_Decr_Oth_Ass): (See **Hint** below)

$$=ACCT.Other_Assets[time=THIS-1]-ACCT.Other_Assets$$

Incr/(Decr) Accounts Payable (CF_Incr_Decr_AP):

$$=ACCT.AP-ACCT.AP[time=THIS-1]$$

Incr/(Decr) Accrued Expense (CF_Incr_Decr_Accr_Exp):

$$=ACCT.Accr_Exp-ACCT.Accr_Exp[time=THIS-1]$$

Incr/(Decr) Deferred Revenue (CF_Incr_Decr_Def_Rev):

$$=ACCT.Def_Rev)-ACCT.Def_Rev[time=THIS-1]$$

Incr/(Decr) Other Liabilities (CF_Incr_Decr_Oth_Liab): (See **Hint** below)

$$= ACCT.Other_Liab -ACCT.Other_Liab[time=THIS-1]$$

Cash Flow from Investing

Purchase of Capital (CF_Purch_Capital) should be driven by the change in all fixed asset accounts (do not include accumulated depreciation; this is accounted for separately, in Net Income Adjustments.)

$$=ACCT.FA_Total[time=THIS-1]-ACCT.FA_Total$$

Cash Flow from Financing

These accounts should be driven by the change in long-term liabilities such as Notes Payable or Capital Leases. Several accounts can be aggregated into one formula, for example:

Capital Lease Change (CF_Chg_Cap_Lease):

$$\begin{aligned} &=((ACCT.Cap.Lease.A-ACCT.Cap.Lease.A[time=THIS-1]) \\ &+(ACCT.Cap.Lease.B - ACCT.Cap.Lease.B[time=THIS-1])) \end{aligned}$$

Adaptive Planning, Inc. Balance Sheet/Cash Flow Guide

Troubleshooting the Cash Flow Statement

It can be difficult to construct a perfect cash flow statement in one effort. Invariably, the “Cash Check” row at the bottom of the sheet displays a value other than zero in the early stages of building the statement. This probably means that something on the balance sheet is not being included on the cash flow statement, or that one or more signs are reversed in cash flow formulas.

One way to check the two sheets against one another is to send the balance sheet to Excel, by clicking on Printable View. Delete all time period columns except for two consecutive plan months, e.g. January and February. Add a new column called “Delta.” Write a formula in all rows which subtracts February from January. This column displays the change in all balance sheet accounts.

6				
7	Account Name	FY2008		
8		Q1		
9		Jan	Feb	Delta
10	Assets			
11	Current Assets			
12	Cash	5,225,779	8,565,596	3,339,817
13	Accounts Receivable	11,251,285	11,277,702	26,417
14	Prepaid Expenses	-109,000	-136,250	-27,250
15	Other Current Assets			
16	Total	16,368,064	19,707,048	3,338,984
17	Fixed Assets			
18	Equipment	0	0	0
19	Hardware	104,030	130,986	26,955
20	Leasehold Improvements	0	0	0
21	Accumulated Depreciation	-14,966	-21,321	-6,355
22	Other Fixed Assets			
23	Total	89,064	109,665	20,600
24	Other Assets			
25	Total	16,457,128	19,816,713	3,359,584
26	Liabilities and Equities			
27	Liabilities			
28	Current Liabilities			
29	Accounts Payable	307,601	307,964	363
30	Deferred Revenue	15,844,913	19,204,130	3,359,217
31	Accrued Expenses	304,614	304,619	5

Compare the values in this column to the values on the cash flow statement for February. If there is a value in this column which is not included somewhere on the cash flow statement, this is the problem. If the Balance sheet balances, then this method should uncover the problem. If the balance sheet is out of balance, that problem should be solved first.

Adaptive Planning, Inc. Balance Sheet/Cash Flow Guide

8. Appendices

Appendix A: Cash Flow Account Import

Required				
Name	Rolls Up To	Type	Code	Time Ro
Cash Flow	Custom			
Cash Flow from All Sources	Cash Flow	Custom (periodic)	CF_from_All_Sources	
Cash Flow from Operating Activities	Cash Flow from All Sources	Custom (periodic)	CF_from_Ops	
Net Income (Loss)	Cash Flow from Operating Activities	Custom (periodic)	CF_NI	
Net Income Adjustments	Cash Flow from Operating Activities	Custom (periodic)	CF_NI_Adj	
Depreciation & Amortization	Net Income Adjustments	Custom (periodic)	CF_Deprec_Amort	
Deferred Taxes	Net Income Adjustments	Custom (periodic)	CF_Def_Tx	
Change in Operating Assets & Liabilities	Cash Flow from Operating Activities	Custom (periodic)	CF_Chg_Oper_A_L	
(Incr)/Decr Accounts Receivable	Change in Operating Assets & Liabilities	Custom (periodic)	CR_Incr_Decr_AR	
(Incr)/Decr in Unbilled Earned Revenue	Change in Operating Assets & Liabilities	Custom (periodic)	CF_Incr_Unb_Earn_Rev	
(Incr)/Decr in Ppd Insurance & Other	Change in Operating Assets & Liabilities	Custom (periodic)	CF_Incr_Decr_PPD_Ins	
Incr/(Decr) in Accounts Payable	Change in Operating Assets & Liabilities	Custom (periodic)	CF_Incr_Decr_AP	
Incr/(Decr) in Accr Expenses & Other	Change in Operating Assets & Liabilities	Custom (periodic)	CF_Incr_Decr_Accr_Exp	
Incr/(Decr) in Deferred Revenue	Change in Operating Assets & Liabilities	Custom (periodic)	CF_Incr_Decr_Def_Rev	
Cash Flow from Investing Activities	Cash Flow from All Sources	Custom (periodic)	CF_from_Investing	
Capital Expenditures	Cash Flow from Investing Activities	Custom (periodic)	CF_Cap_Exp	
Change in Deposits & Other	Cash Flow from Investing Activities	Custom (periodic)	CF_Incr_Chg	
Cash Flow from Financing Activities	Cash Flow from All Sources	Custom (periodic)	CF_from_Fin	
Change in Capital Leases	Cash Flow from Financing Activities	Custom (periodic)	CF_Cap_Ls_Chg	
Change in Capital Stock	Cash Flow from Financing Activities	Custom (periodic)	CF_Cap_Stock_Chg	
Additional Paid-In Capital	Cash Flow from Financing Activities	Custom (periodic)	CF_Addl_Paid_In_Cap	
Distribution to Stockholders	Cash Flow from Financing Activities	Custom (periodic)	CF_Dist_to_Stock	
Change in Cash (Components)	Cash Flow			
Beginning Cash & Equivalents	Change in Cash (Components)	Custom (cumulative)	CF_Beg_Bal	
Change in Cash	Change in Cash (Components)	Custom (periodic)	CF_Chg	
Ending Cash & Equivalents	Change in Cash (Components)	Custom (cumulative)	CF_End_Bal	
Cash Check	Cash Flow			
Cash Balance on Balance Sheet	Cash Check	Custom (cumulative)	CF_Cash_from_BS	
Cash Check (s/b zero)	Cash Check	Custom (periodic)	CF_Check	

Click here to download the Excel file:

<http://www.adaptiveplanning.com/docs/Training/CashFlowAccountImport.xls>

Adaptive Planning, Inc. Balance Sheet/Cash Flow Guide

Appendix B: CTA Calculation Example

Exchange rate	1	1	1.1	1.2	1.3	1.4	1.25	1	0.9	0.9	1	0.85	
	Dec	Jan	Feb	Mar	Apr	May	Jun	Jul	Aug	Sep	Oct	Nov	Dec
Plan: CAD (Local)													
Common Stock	0	5,000,000	5,000,000	5,000,000	5,000,000	5,000,000	5,000,000	5,000,000	5,000,000	5,000,000	7,000,000	7,000,000	7,000,000
Plan: CAD rolled up to USD w/o CTA (converted automatically - which is not the desired behavior)													
Common Stock	5,000,000	5,000,000	5,500,000	6,000,000	6,500,000	7,000,000	6,250,000	5,000,000	4,500,000	6,300,000	7,000,000	5,950,000	
Plan: Elimination in USD (new plan)													
Gain/Loss			-	500,000	1,000,000	1,500,000	2,000,000	1,250,000	-	(500,000)	(500,000)	200,000	(850,000)
Common Stock		-	-	(500,000)	(1,000,000)	(1,500,000)	(2,000,000)	(1,250,000)	-	500,000	500,000	(200,000)	850,000
End Result Consolidated (rollup of new plan + CAD converted as desired)													
Gain/Loss			-	500,000	1,000,000	1,500,000	2,000,000	1,250,000	-	(500,000)	(500,000)	200,000	(850,000)
Common Stock		5,000,000	5,000,000	5,000,000	5,000,000	5,000,000	5,000,000	5,000,000	5,000,000	5,000,000	6,800,000	6,800,000	6,800,000

Key - in the month of activity, the change (2M in October) needs to be converted at the current rate and show the same conversion going forward. If the rate changes, that change should hit a Gain Loss Account. In this example the 5M in January (at 1:1) and the 2M in October (@ 0.9:1) should reflect the same converted amount in every month

Click here to download the Excel file:

<http://www.adaptiveplanning.com/docs/Training/CTAExample.xls>